Formal Proof of Type Preservation of the Dictionary Passing Transform for System F

Marius Weidner

Chair of Programming Languages, University of Freiburg
weidner@cs.uni-freiburg.de

Bachelor Thesis

Examiner: Prof. Dr. Peter Thiemann
Advisor: Hannes Saffrich

Abstract. Most popular strongly typed programming languages support function overloading. In combination with polymorphism this leads to essential language constructs, for example typeclasses in Haskell or traits in Rust. We introduce System F_O, a minimal language extension to System F, with support for overloading and polymorphism. Furthermore, we prove the Dictionary Passing Transform from System F_O to System F to be type preserving using Agda.

Abstract. (german) Viele stark getypte Programmiersprachen unterstützen das Überladen von Funktionsnamen. In Kombination mit Polymorphismus ergeben sich essenzielle Sprachkonstrukte, beispielsweise Typklassen in Haskell, oder Traits in Rust. Wir erarbeiten System F_O, eine minimale Spracherweiterung von System F, die Überladung und Polymorphismus unterstützt. Anschließend beweisen wir formal in Agda, dass die Dictionary Passing Transform von System F_O zu System F typerhaltend ist.

Table of Contents

1 Introduction

1.1 Overloading in Programming Languages

Overloading function names is a practical technique to overcome verbosity in real world programming languages. In every language there exist commonly used function names and operators that are defined for a variety of type combinations. Overloading the meaning of a function name helps to solve the problem of having to define similar but differing names and operators for different type combinations. Overloading is also sometimes referred to as ad-hoc polymorphism. An ad-hoc polymorphic function is allowed to have multiple type-specific meanings for all types that it is defined for. In contrast, a parametric polymorphic function only defines abstract behavior that must work for all types.
Python, for example, uses magic methods to overload commonly used operators on user defined classes and Java utilizes method overloading. Both Python and Java implement rather restricted forms of overloading. Haskell solves the overloading problem with a more general concept, called typeclasses.

1.2 Typeclasses in Haskell

Essentially, typeclasses allow to declare function names with generic type signatures. We can give one of possibly many meanings to a typeclass by instantiating the typeclass for concrete types. When we instantiate a typeclass, we must provide an actual implementation for all functions defined by the typeclass, based on the concrete types that the typeclass is instantiated for. When we invoke an overloaded function name defined by a typeclass, we expect the compiler to determine the correct instance, based on the types of the arguments that were applied to the overloaded function name. Furthermore, Haskell allows to constrain type variables via type constraints to only be substituted by a concrete type, if there exists an instance for the concrete type. Type constraints allow to abstract over all types that inherit a shared behavior, but the actual implementation of the behavior can differ per type. Type constraints are a powerful formalism in addition to parametric polymorphism.

Example: Overloading Equality in Haskell

In this example the function `eq : α → α → Bool` is overloaded with different meanings for different substitutions $\{\alpha \mapsto \tau\}$. We want to be able to call `eq` on both $\{\alpha \mapsto \texttt{Nat}\}$ and $\{\alpha \mapsto \texttt{[}\beta\texttt{]}\}$, where β is a type and there exists an instance that gives meaning to `eq : β → β → Bool`. The intuition here is that we want to be able to compare natural numbers `Nat` and lists `[β]`, given the elements of type β are known to be comparable.

```
class Eq α where
  eq :: α → α → Bool

instance Eq Nat where
  eq x y = x ≐ y
instance Eq β ⇒ Eq [β] where
  eq []       []       = True
  eq (x : xs) (y : ys) = eq x y && eq xs ys

.. eq 42 0 .. eq [42, 0] [42, 0] ..
```

First, typeclass `Eq` is declared with a single generic function signature `eq :: α → α → Bool`. We then instantiate `Eq` for $\{\alpha \mapsto \texttt{Nat}\}$. After that, `Eq` is instantiated for $\{\alpha \mapsto [\beta]\}$, given that an instance `Eq β` can be resolved for some concrete type `β`. As a result, we can invoke `eq` on expressions with type `Nat` and `[Nat]`. In the latter case, the type constraint `Eq β ⇒ ..` in the instance for lists resolves to the instance for natural numbers.

1.3 Desugaring Typeclass Functionality to System F_O

System F_O is a minimal calculus with support for overloading and polymorphism based on System F. System F is also sometimes referred to as polymorphic lambda calculus or second-order lambda calculus. In System F_O we give up high level language constructs and instead work with simple overloaded identifiers.
Using the `decl o in e` expression we can introduce an new overloaded variable `o`. If declared as overloaded, `o` can be instantiated for the type `τ` of the expression `e` using the `inst o = e in e'` expression. In Haskell instances must comply with the generic type signatures defined by the typeclass. Such signatures are not present in System F_O and overloaded variables can be instantiated for arbitrary types. Locally shadowing other instances of the same type is allowed. Constraints can be introduced on the expression level using constraint abstractions `λ (o : τ). e'`. A constraint `o : τ` requires the type system to search for an instance for the overloaded variable `o` that has type `τ`. If the constraint cannot be resolved, then the program is invalid. Constraint abstractions result in constraint types `[o : τ] ⇒ τ'` that lift constraints onto the type level. We introduce constraints on the expression level because instance expressions do not have an explicit type annotation in System F_O. All expressions with constraint types `[o : τ] ⇒ τ'` are implicitly treated as expressions of type `τ'` by the type system, given that the constraint `o : τ` can be resolved.

Example: Overloading Equality in System F_O

Recall the Haskell example from above. The same functionality can be expressed in System F_O. For convenience, type annotations for instances are given.

```
decl eq in

inst eq : Nat → Nat → Bool
  = λx. λy. .. in
inst eq : ∀β. [eq : β → β → Bool] ⇒ [β] → [β] → Bool
  = Λβ. λ(eq : β → β → Bool). λxs. λys. .. in

.. eq 42 0 .. eq Nat [42, 0] [42, 0] ..
```

We first declare `eq` to be an overloaded identifier and instantiate `eq` for equality on `Nat`. Next, we instantiate `eq` for equality on lists `[β]`, given that the constraint `eq : β → β → Bool` introduced by the constraint abstraction is satisfied. Because System F_O is based on System F, a calculus without type inference, we are required to bind type variables using type abstractions `Λα. ..` and eliminate type variables using type application.
A little caveat: the instance for lists would potentially need to recursively call `eq` for sublists, but the formalization of System F_O does not actually support recursion. Extending System F_O with recursive let bindings and thus recursive instances is known to be straight forward.

1.4 Translating System F_O back to System F

System F_O can be translated back to System F. This implies that System F_O is no more expressive or powerful than System F. After all, overloading is more of a convenience feature.
The Dictionary Passing Transform translates well typed System F_O expressions to well typed System F expressions. The translation requires knowledge acquired during type checking. More specifically, we need to know the instances that were resolved for invocations of overloaded identifiers and the instances that constraints were implicitly resolved with.
The translation removes all `decl` expressions. Instance expressions `inst o = e in e'` are replaced with `let` o_τ `= e in e'` expressions, where o_τ is a unique name with respect to the type `τ` of the expression `e`. Implicitly resolved constraints in System F_O will be taken as explicit higher-order function arguments in System F. As a result, constraint abstractions `λ (o : τ). e'` translate to normal abstractions `λ`o_τ`. e'` and constraint types `[o : τ] ⇒ τ'` translate to function types `τ → τ'`. A invocation of an overloaded function name `o` translates to the correct unique variable name o_τ, that is bound by the let binding that got introduced for the corresponding resolved instance. The translation becomes more intuitive when looking at an example.

Example: Dicitionary Passing Transform

Recall the System F_O example from above. We use indices to represent unique names o_τ. Applying the Dictionary Passing Transform to the example above results in a well typed System F expression. Type annotations for let bindings are given for convenience.

```
let eq₁ : Nat → Nat → Bool
  = λx. λy. .. in
let eq₂ : ∀β. (β → β → Bool) → [β] → [β] → Bool
  = Λβ. λeq₁. λxs. λys. .. in

.. eq₁ 42 0 .. eq₂ Nat eq₁ [42, 0] [42, 0] ..
```

We remove the `decl` expression and transform both `inst` expressions to `let` bindings with unique names $\texttt{eq}_i$. Inside the instance for lists, the constraint abstraction translates to a normal lambda abstraction. The lambda abstraction takes the constraint that was implicitly resolved in System F_O as an explicit higher-order function argument. Invocations of `eq` translate to correct unique variables $\texttt{eq}_i$ that are bound by the let bindings that got introduced for the former resolved instances. Because $\texttt{eq}_2$ is invoked for a list of numbers, we must pass the correct instance to eliminate the new higher-order function binding by explicitly passing instance $\texttt{eq}_1$ as argument.

2 Preliminary

2.1 Dependently Typed Programming in Agda

Agda is a dependently typed programming language and proof assistant [3]. Agda's type system is based on intuitionistic type theory and allows to construct proofs based on the Curry-Howard correspondence. The Curry-Howard correspondence is an isomorphic relationship between programs written in dependently typed programming languages and mathematical proofs written in first order logic. Because of the Curry-Howard correspondence, programs correspond to proofs and formulae correspond to types. Thus, type checked Agda programs imply the correctness of the corresponding proofs, assuming that we do not use unsafe Agda features and Agda is implemented correctly. We will use Agda to formalize the type preservation proof for the Dictionary Passing Transform from System F_O to System F.

2.2 Design Decisions for the Agda Formalization

To formalize the syntaxes of System F and System F_O in Agda we use a single data type `Term` indexed by sorts s. Sorts distinguish between different categories of terms. For example, the sort $\texttt{e}_s$ represents expressions e, $\texttt{τ}_s$ represents types τ and $\texttt{κ}_s$ represents kinds. In System F and System F_O there only exists a single kind $\star$. The name 'sort' originates from the theory of pure type systems [2], but neither System F nor System F_O allow any interesting dependencies between terms of sort $\texttt{e}_s$, $\texttt{τ}_s$, and $\texttt{κ}_s$. Using a single data type to formalize syntaxes yields more elegant proofs involving contexts, renamings and substitutions. In consequence of using a single data type, we must use extrinsic typing because intrinsically typed terms would need to be indexed by themselves and Agda does not support self-indexed data types. In the actual implementation, the data type `Term` has another index S that we will ignore for now.

2.3 Overview of the Type Preservation Proof

The overall goal will be to prove that the Dictionary Passing Transform is type preserving. Let $\vdash t$ be any well formed System F_O term $\Gamma \vdash_{F_O} t : T$, where Γ is a typing

context of type Ctx_{F_O}, t is a $\mathsf{Term}_{F_O}\ s$, T is a $\mathsf{Term}_{F_O}\ s'$ and s' is the sort of the typing result for terms of the sort s. There are two cases for typings: $\Gamma \vdash e : \tau$ and $\Gamma \vdash \tau : \star$. Let $\rightsquigarrow : (\Gamma \vdash_{F_O} t : T) \to \mathsf{Term}_F\ s$ be the Dictionary Passing Transform that translates well typed System F_O terms to untyped System F terms. Further, let $\rightsquigarrow_\Gamma : \mathsf{Ctx}_{F_O} \to \mathsf{Ctx}_F$ be the transform of contexts and $\rightsquigarrow_T : \mathsf{Term}_{F_O}\ s' \to \mathsf{Term}_F\ s'$ be the transform of untyped types and kinds. We show that for all well typed System F_O terms $\vdash t$ the Dictionary Passing Transform results in a well typed System F term $(\rightsquigarrow_\Gamma\ \Gamma) \vdash_F (\rightsquigarrow \vdash t) : (\rightsquigarrow_T\ T)$.
We begin by formalizing the syntax, typing and semantic of System F and prove its soundness in section 3. In section 4, we formalize System F_O's syntax and typing. In the end, we formalize the translation of the Dictionary Passing Transform and prove it to be type preserving in section 5.
We do not formalize semantics and soundness for System F_O. In a way, correct semantics for System F_O are already given by the type preserving translation from System F_O to System F. This is because we can simply apply the semantics of System F after translating a given System F_O program. And furthermore, the semantics of System F are proven to be sound in combination with the type system that System F_O is safely translated to.

3 System F

3.1 Specification

Sorts

The formalization of System F requires three sorts: e_s for expressions, τ_s for types and κ_s for kinds.

```
data Sort : Bindable → Set where
  eₛ : Sort var
  τₛ : Sort var
  κₛ : Sort no-var
```

Sorts are indexed by the boolish data type Bindable. The index var indicates that variables for terms of a sort can be bound. In contrast, the index no-var says that variables for terms of a sort cannot be bound. In this case, System F supports abstracting over expressions and types, but not over kinds. Going forward, we will use the variable r for elements of type Bindable, the variable s for elements of type Sort and the variable S for lists of bindable sorts with type Sorts = List (Sort var).

Syntax

The syntax of System F is represented in a single data type Term that is indexed by sorts S and sort s. The index S is inspired by Debruijn indices. Debruijn indices reference variables using a number that counts the amount of binders that are in scope between the binding of the variable and the position it is used at. In Agda terms are often indexed by the amount of bound variables. The variable constructor then only accepts Debruijn indices, instead of variable names, that are smaller than the current amount of bound variables. As a result, unbound variables cannot be referenced by

definition. This technique is also referred to as intrinsically scoped. But indexing Term with a number is not sufficient because System F has both expression variables and type variables, that need to be distinguished. To solve this problem, we need to extend the idea of Debruijn indices and store the corresponding sort for each variable. Thus, we let S be a list of bindable sorts, instead of a number. The length of S represents the amount of bound variables and the elements s_i of the list represent the sort of the variable bound at Debruijn index i.
The index s represents the sort of the term itself.

```
data Term : Sorts → Sort r → Set where
  ‘_            : s ∈ S → Term S s
  tt            : Term S e_s
  λ‘x→_         : Term (S ▷ e_s) e_s → Term S e_s
  Λ‘α→_         : Term (S ▷ τ_s) e_s → Term S e_s
  _·_           : Term S e_s → Term S e_s → Term S e_s
  _•_           : Term S e_s → Term S τ_s → Term S e_s
  let‘x=_‘in_   : Term S e_s → Term (S ▷ e_s) e_s → Term S e_s
  ‘⊤            : Term S τ_s
  _⇒_           : Term S τ_s → Term S τ_s → Term S τ_s
  ∀‘α_          : Term (S ▷ τ_s) τ_s → Term S τ_s
  ⋆             : Term S κ_s
```

Variables ‘ x are represented as membership proofs of type $s \in S$. Membership proofs are inductively defined, similar to the definition of natural numbers. Membership proofs can be constructed using the constructor here refl, where refl is proof that the last element in a list S is the element we searched for. Alternatively, membership proofs for a list S can be constructed via the constructor there x, where x is a membership proof for the sublist S' of S that has one element less. As discussed, the Debruijn representation of variables has the advantage that only already bound variables can be referenced by the variable constructor by definition.
The unit element tt and type ‘⊤ represent base expressions and types respectively.
Lambda abstractions λ‘x→ e result in function types $\tau_1 \Rightarrow \tau_2$ and type abstractions Λ‘α→ e result in forall types ∀‘α τ. Both abstractions and forall types introduce an additional sort e_s, or τ_s respectively, to the index S of their corresponding body to account for the additional new binding.
The application constructor $e_1 \cdot e_2$ applies the argument e_2 to the function e_1.
Similarly, type application $e \bullet \tau$ eliminates type abstractions.
Let bindings let‘x= e_2 ‘in e_1 combine abstraction and application.
The kind ⋆ is kind of all types.
We use abbreviations Var S s = $s \in S$, Expr S = Term S e_s, Type S = Term S τ_s and variables x, e and τ respectively. Furthermore, we use the variable t for an arbitrary Term S s.

Renaming

Renamings ρ of type Ren S_1 S_2 are defined as total functions that map variables Var S_1 s to variables Var S_2 s. Renamings preserve the sort s of the variable.

```
Ren : Sorts → Sorts → Set
Ren S₁ S₂ = ∀ s → Var S₁ s → Var S₂ s
```

Applying a renaming Ren S_1 S_2 to a term Term S_1 s yields a new term Term S_2 s, where variables are represented as references to elements in S_2 instead of S_1. The function ren applies a renaming to a term.

```
ren : Ren S₁ S₂ → (Term S₁ s → Term S₂ s)
ren ρ (` x) = ` (ρ _ x)
ren ρ (λ`x→ e) = λ`x→ (ren (extᵣ ρ _) e)
ren ρ (τ₁ ⇒ τ₂) = ren ρ τ₁ ⇒ ren ρ τ₂
- ...
```

In the first case, the renaming is applied to all variables x.
When we encounter a binder for a term of sort s, as seen in the second case, the renaming is extended using function ext_r. If we want to use a renaming as a function or use the function ext_r, the sort argument s can usually be inferred by Agda. Inferring a function argument is denoted with _.

```
extᵣ : Ren S₁ S₂ → (s : Sort var) → Ren (S₁ ▷ s) (S₂ ▷ s)
extᵣ ρ _ _ (here refl) = here refl
extᵣ ρ _ _ (there x) = there (ρ _ x)
```

The extension of a renaming introduces an additional variable binding of sort s. Thus, if we encounter the new binding here refl in the extended renaming, then we return the variable for the new binding here refl. The variables x of the original renaming ρ are weakened by wrapping them in an additional there constructor. The sort arguments are ignored inside the function body of ext_r by using the wildcard pattern _.
Similar to variables, terms can be weakened using the function wk that shifts all variables present in the term by one recursively.

```
wk : Term S s → Term (S ▷ s') s
wk = ren wkᵣ
```

The function wk_r generates a weakening by wrapping all variables in an additional there constructor.

```
wkᵣ : Ren S (S ▷ s)
wkᵣ _ = there
```

Substitution

The definition of substitutions σ with type Sub S_1 S_2 is similar to the definition of renamings. But rather than mapping variables to variables, substitutions map variables to terms.

```
Sub : Sorts → Sorts → Set
Sub S₁ S₂ = ∀ s → Var S₁ s → Term S₂ s
```

Applying a substitution using the sub function is analogous to applying a renaming using ren. If we encounter a binder in sub, the substitution must be extended using function ext_s.

```
extₛ : Sub S₁ S₂ → (s : Sort var) → Sub (S₁ ▷ s) (S₂ ▷ s)
extₛ σ s _ (here refl) = ` here refl
extₛ σ s _ (there x) = wk (σ _ x)
```

For the newly bound variable here refl, we return the variable term ‘ here refl. Furthermore, all terms $\sigma\ \_\ x$ originally present in the substitution σ are weakened using the function wk.
The substitution operator $t\ [\ t'\ ]$ substitutes the last bound variable in t with t', given that the sort of the last binder corresponds to the sort of t'.

```
_[_] : Term (S ▷ s') s → Term S s' → Term S s
t [ t' ] = sub (single_s id_s t') t
```

A single substitution single_s takes an existing substitution σ' and term t'. The term t' is then introduced for an additional new binding here refl.

```
single_s : Sub S₁ S₂ → Term S₂ s → Sub (S₁ ▷ s) S₂
single_s σ t' _ (here refl) = t'
single_s σ t' _ (there x) = σ _ x
```

In the case of _[_], we let σ' be the identity substitution id_s : Sub S S.

Context

Similar to terms, typing contexts Γ are indexed by a list of bound variables. In consequence, only types and kinds for bound variables can be stored in Γ by definition.

```
data Ctx : Sorts → Set where
  ∅ : Ctx []
  _►_ : Ctx S → Term S (type-of s) → Ctx (S ▷ s)
```

Contexts are inductively defined and can either be empty ∅ or extended with one element T, using the constructor $\Gamma \blacktriangleright T$. The variable T represents terms of the sort type-of s. The function type-of maps bindable sorts s to the sort of the term that is stored in Γ for variables of the sort s. Thus, if we bind a new variable for a term of the sort s, then Γ needs to be extended by a term of the sort type-of s.

```
type-of e_s = τ_s
type-of τ_s = κ_s
```

Expression variables require Γ to store the corresponding type. Similarly, we store the corresponding kind for all types in Γ.
The lookup function resolves the type or kind T for a variable in the context Γ.

```
lookup : Ctx S → Var S s → Term S (type-of s)
lookup (Γ ► T) (here refl) = wk T
lookup (Γ ► T) (there x) = wk (lookup Γ x)
```

Inside both cases of the case split on variables, we wrap the looked up T in a weakening. As a result, T always has the index S that aligns with the current required amount of bound variables. The lookup function cannot fail by definition because we only allow to lookup bound variables that must have an entry in Γ by definition.

Typing

The typing relation $\Gamma \vdash t : T$ relates a term t to its typing result T in a context Γ.

```
data _⊢_:_ : Ctx S → Term S s → Term S (type-of s) → Set where
  ⊢`x :
    lookup Γ x ≡ τ →
    Γ ⊢ ` x : τ
  ⊢⊤ :
    Γ ⊢ tt : `⊤
  ⊢λ :
    Γ ▶ τ ⊢ e : wk τ' →
    Γ ⊢ λ`x→ e : τ ⇒ τ'
  ⊢Λ :
    Γ ▶ ⋆ ⊢ e : τ →
    Γ ⊢ Λ`α→ e : ∀`α τ
  ⊢· :
    Γ ⊢ e₁ : τ₁ ⇒ τ₂ →
    Γ ⊢ e₂ : τ₁ →
    Γ ⊢ e₁ · e₂ : τ₂
  ⊢• :
    Γ ⊢ e : ∀`α τ →
    Γ ⊢ e • τ' : τ [ τ' ]
  ⊢let :
    Γ ⊢ e₂ : τ →
    Γ ▶ τ ⊢ e₁ : wk τ' →
    Γ ⊢ let`x= e₂ `in e₁ : τ'
  ⊢τ :
    Γ ⊢ τ : ⋆
```

The rule ⊢`x says that a variable ` x has type τ, if the type for x in Γ is τ.
All unit expressions tt have type `⊤. This is expressed by the rule ⊢⊤.
The rule for abstractions ⊢λ introduces an expression variable of type τ to the body e. Because the resulting body type τ' cannot use the newly introduced expression variable, we let τ' have one variable bound less and weaken it to align in the list of bound variables with the context $\Gamma \blacktriangleright \tau$. As a result, τ' aligns with τ in the list of bound variables to form the resulting function type $\tau \Rightarrow \tau'$.
The type abstraction rule ⊢Λ introduces a type variable to the body e and results in the forall type ∀`α τ, where τ is the type of e. The type variable in e is introduced by extending Γ with the kind ⋆.
Application is handled by the rule ⊢·. The rule says that if e_1 is a function from τ_1 to τ_2 and e_2 has type τ_1, then $e_1 \cdot e_2$ has type τ_2.
Similarly, the type application rule ⊢• states that if e has type ∀`α τ, then a can be substituted with another type τ' inside τ.
The rule ⊢let combines the abstraction and application rule.
Regarding the typing of types, the rule ⊢τ indicates that all types τ are well formed and have kind ⋆. Type variables are correctly typed per definition and type constructors ∀`α and ⇒ accept arbitrary types as their arguments. Hence, all types are well typed.

Typing of Renaming & Substitution

Because of extrinsic typing, both renamings and substitutions need to have typed counterparts.

We formalize typed renamings $\vdash\rho$ inductively as order preserving embeddings. Thus, if a variable x_1 of type $s_1 \in S_1$ references an element with an index smaller than some other variable x_2 in S_1, then renamed x_1 must still reference an element with a smaller index than renamed x_2 in S_2. Arbitrary renamings would allow swapping items and potentially violate the telescoping. Telescoping allows types stored in the context to depend on type variables bound inside the context before them.
Interestingly, because of the intrinsically scoped definition of terms, all renamings must be order preserving embeddings by definition. Thus, it should be possible to prove order preservation in the form of lemmas. Instead we choose to represent the rules for order preserving embeddings as constructors of a data type, such that we can access the property of order preservation by matching on the data type.

```
data _:_⇒ᵣ_ : Ren S₁ S₂ → Ctx S₁ → Ctx S₂ → Set where
  ⊢idᵣ : ∀ {Γ} → _:_⇒ᵣ_ {S₁ = S} {S₂ = S} idᵣ Γ Γ
  ⊢extᵣ : ∀ {ρ : Ren S₁ S₂} {Γ₁ : Ctx S₁} {Γ₂ : Ctx S₂}
            {T' : Term S₁ (type-of s)} →
    ρ : Γ₁ ⇒ᵣ Γ₂ →
    (extᵣ ρ _) : (Γ₁ ▶ T') ⇒ᵣ (Γ₂ ▶ ren ρ T')
  ⊢dropᵣ : ∀ {ρ : Ren S₁ S₂} {Γ₁ : Ctx S₁} {Γ₂ : Ctx S₂}
             {T' : Term S₂ (type-of s)} →
    ρ : Γ₁  ⇒ᵣ Γ₂ →
    (dropᵣ ρ) : Γ₁ ⇒ᵣ (Γ₂ ▶ T')
```

The identity renaming $\vdash\mathsf{id}_r$ is typed by definition.
The typed extension of a renaming $\vdash\mathsf{ext}_r$ allows to extend both Γ_1 and Γ_2 by T' and renamed T' respectively. The constructor $\vdash\mathsf{ext}_r$ corresponds to the typed version of the function ext_r that is used when a binder is encountered.
The constructor $\vdash\mathsf{drop}_r$ allows to introduce T' only in Γ_2. Hence, $\vdash\mathsf{drop}_r\ \vdash\mathsf{id}_r$ corresponds to the typed weakening $\vdash\mathsf{wk}_r$ of a term.
The absence of a constructor that allows to introduce some T only to Γ_1 is exactly the restriction needed for typed renamings to be order preserving embeddings.
Typed Substitutions are defined as total functions, similar to untyped substitutions.

```
_:_⇒ₛ_ : Sub S₁ S₂ → Ctx S₁ → Ctx S₂ → Set
_:_⇒ₛ_ {S₁ = S₁} σ Γ₁ Γ₂ = ∀ {s} (x : Var S₁ s) →
                              Γ₂ ⊢ σ _ x : (sub σ (lookup Γ₁ x))
```

Typed substitutions $\vdash\sigma$ map variables $x \in S_1$ to the corresponding typing of the term $\sigma\ \_\ x$ in Γ_2. The type of the term $\sigma\ \_\ x$ must be the original type of x in Γ_1 applied to the substitution σ.

Semantics

The semantics of System F are formalized as call-by-value, that is, there is no reduction under binders.
Values are indexed by their corresponding irreducible expression.

```
data Val : Expr S → Set where
  v-λ : Val (λ'x→ e)
  v-Λ : Val (Λ'α→ e)
  v-tt : ∀ {S} → Val (tt {S = S})
```

System F has three values. The two closure values v-λ and v-Λ and the unit value v-tt. We formalize small step semantics, where each constructor represents a single reduction step $e \hookrightarrow e'$. Small step semantics distinguish between β and ξ rules. Meaningful computation in the form of substitution is done by β rules while ξ rules only reduce subexpressions.

```
data _↪_ : Expr S → Expr S → Set where
  β-λ :
    Val e₂ →
    (λ'x→ e₁) · e₂ ↪ e₁ [ e₂ ]
  β-Λ :
    (Λ'α→ e) • τ ↪ e [ τ ]
  β-let :
    Val e₂ →
    let'x= e₂ 'in e₁ ↪ (e₁ [ e₂ ])
  ξ-·₁ :
    e₁ ↪ e →
    e₁ · e₂ ↪ e · e₂
  ξ-·₂ :
    e₂ ↪ e →
    Val e₁ →
    e₁ · e₂ ↪ e₁ · e
  ξ-• :
    e ↪ e' →
    e • τ ↪ e' • τ
  ξ-let :
    e₂ ↪ e →
    let'x= e₂ 'in e₁ ↪ let'x= e 'in e₁
```

The rules β-λ and β-Λ give meaning to application and type application by substituting the applied expression, or type respectively, into the abstraction body. In both cases, we make sure that the abstraction and the applied argument are values.
The reduction rule β-let functions similar to the rule β-λ and substitutes the value e_2 into e_1.
The rules ξ-$\cdot_i$ and ξ-• evaluate subexpressions of applications until e_1 and e_2, or e respectively, are values.
The rule ξ-let reduces the bound expression e_2 until e_2 is a value and β-let can be applied.

3.2 Soundness

Progress

We prove progress by showing that a typed expression e can either be further reduced to another expression e' or e is a value. The proof follows by induction over the typing rules.

```
progress :
  ∅ ⊢ e : τ →
  (∃[ e' ] (e ↪ e')) ⊎ Val e
progress ⊢⊤ = inj₂ v-tt
progress (⊢λ _) = inj₂ v-λ
progress (⊢Λ _) = inj₂ v-Λ
progress (⊢· {e₁ = e₁} {e₂ = e₂} ⊢e₁ ⊢e₂) with progress ⊢e₁ | progress ⊢e₂
... | inj₁ (e₁' , e₁↪e₁') | _ = inj₁ (e₁' · e₂ , ξ-·₁ e₁↪e₁')
... | inj₂ v | inj₁ (e₂' , e₂↪e₂') = inj₁ (e₁ · e₂' , ξ-·₂ e₂↪e₂' v)
... | inj₂ (v-λ {e = e₁}) | inj₂ v = inj₁ (e₁ [ e₂ ] , β-λ v)
progress (⊢• {τ' = τ'} ⊢e) with progress ⊢e
... | inj₁ (e' , e↪e') = inj₁ (e' • τ' , ξ-• e↪e')
... | inj₂ (v-Λ {e = e}) = inj₁ (e [ τ' ] , β-Λ)
progress (⊢let {e₂ = e₂} {e₁ = e₁} ⊢e₂ ⊢e₁) with progress ⊢e₂
... | inj₁ (e₂' , e₂↪e₂') = inj₁ ((let'x= e₂' 'in e₁) , ξ-let e₂↪e₂')
... | inj₂ v = inj₁ (e₁ [ e₂ ] , β-let v)
```

The cases ⊢⊤, ⊢λ and ⊢Λ result in values. The application cases ⊢·, ⊢• and ⊢let follow directly from the induction hypothesis.

Subject Reduction

We prove subject-reduction, that is, reduction preserves typing. More specifically, an expression e with type τ still has type τ after being reduced to e'. We prove subject reduction by induction over the reduction rules.

```
subject-reduction : ∀ {Γ : Ctx S} →
  Γ ⊢ e : τ →
  e ↪ e' →
  Γ ⊢ e' : τ
subject-reduction (⊢· (⊢λ ⊢e₁) ⊢e₂) (β-λ v₂) = e[e]-preserves ⊢e₁ ⊢e₂
subject-reduction (⊢· ⊢e₁ ⊢e₂) (ξ-·₁ e₁↪e) = ⊢· (subject-reduction ⊢e₁ e₁↪e) ⊢e₂
subject-reduction (⊢· ⊢e₁ ⊢e₂) (ξ-·₂ e₂↪e x) = ⊢· ⊢e₁ (subject-reduction ⊢e₂ e₂↪e)
subject-reduction (⊢• (⊢Λ ⊢e)) β-Λ = e[τ]-preserves ⊢e ⊢τ
subject-reduction (⊢• ⊢e) (ξ-• e↪e') = ⊢• (subject-reduction ⊢e e↪e')
subject-reduction (⊢let ⊢e₂ ⊢e₁) (β-let v₂) = e[e]-preserves ⊢e₁ ⊢e₂
subject-reduction (⊢let ⊢e₂ ⊢e₁) (ξ-let e₂↪e') = ⊢let
  (subject-reduction ⊢e₂ e₂↪e') ⊢e₁
```

The ξ reduction cases ξ-·₁, ξ-·₂, ξ-• and ξ-let follow directly from the induction hypothesis.
For the β reduction cases β-λ, β-Λ and β-let, we need to prove that substitutions preserve the typing. We have two different types of substitution present inside the reduction rules: $e\ [\ e\ ]$ and $e\ [\ \tau\ ]$. Both e[e]-preserves and e[τ]-preserves follow from a more general lemma ⊢σ-preserves. The lemma ⊢σ-preserves proves that applying a typed substitution preserves the typing.

```
⊢σ-preserves : ∀ {σ : Sub S₁ S₂} {Γ₁ : Ctx S₁} {Γ₂ : Ctx S₂}
                 {t : Term S₁ s} {T : Term S₁ (type-of s)} →
  σ : Γ₁ ⇒ₛ Γ₂ →
  Γ₁ ⊢ t : T →
  Γ₂ ⊢ (sub σ t) : (sub σ T)
```

The lemma ⊢σ-preserves follows by induction over the typing rules and lemmas about the interaction of substitutions and renamings. More specifically, we also need to prove that all operations on substitutions preserve the typing. For instance, we need to prove the lemma ⊢σ↑ that says that the typed extension of a substitution ⊢ext_s is type preserving. Because ext_s uses renaming under the hood, we also need to prove the lemma ⊢ρ-preserves that says that applying a typed renaming preserves the typing. Furthermore, we need to prove the lemmas assoc-sub-ren, assoc-ren-ren, assoc-ren-sub and assoc-sub-sub that prove the operations of applying a renaming and substitution to be associative in all combinations. [1].
The soundness property of System F follows as a consequence of progress and subject-reduction.

4 System $\mathrm{F_O}$

4.1 Specification

Sorts

In addition to the sorts of System F, System $\mathrm{F_O}$ introduces two new sorts: o_s for overloaded variables and c_s for constraints.

```
data Sort : Bindable → Set where
  oₛ : Sort var
  cₛ : Sort no-var
  - ...
```

Terms of sort o_s can only be constructed using the variable constructor `_. Thus, terms of sort o_s are called overloaded variables and sort o_s is indexed by var. Variables for constraints do not exist in System $\mathrm{F_O}$ and thus c_s is indexed by no-var. We use the variable symbol o for overloaded variables and the variable symbol c for constraints.

Syntax

We only discuss additions to the syntax of System F.

[1] Considering the fact that the soundness proof for System F is not the main part of this work and resources can be found online [5], the overview of the proof itself is rather short. The full proof can be found as Agda code file: `https://github.com/Mari-W/System-Fo/blob/main/proofs/SystemF.agda`

```
data Term : Sorts → Sort r → Set where
  decl‘o‘in_       : Term (S ▷ o_s) e_s → Term S e_s
  inst‘_‘=_‘in_ : Term S o_s → Term S e_s → Term S e_s → Term S e_s
  _:_              : Term S o_s → Term S τ_s → Term S c_s
  λ_⇒_             : Term S c_s → Term S e_s → Term S e_s
  [_]⇒_            : Term S c_s → Term S τ_s → Term S τ_s
  -  ...
```

Declarations decl‘o‘in e introduce a new overloaded variable o. Hence, S is extended by the sort o_s inside the body e.
The expression inst‘ $o = e_2$ ‘in e_1 introduces an additional instance for o. The actual meaning for the instance is given by e_2. Instance expressions do not introduce new bindings and thus, the index S is never extended.
Constraints c can be constructed using constructor $o : \tau$.
A constraint c can be part of a constraint abstraction $\lambda\ c \Rightarrow e$. Constraint abstractions assume the constraint c to be valid inside the body e and result in constraint types $[\ c\ ]\Rightarrow\ \tau$. The constraint type lifts the constraint from the expression level to the type level, where it will be implicitly eliminated by the typing rules.
Going forward, we will use the abbreviation Cstr S = Term S c_s.

Renaming & Substitution

Renamings and substitutions in System F_O are formalized identically to renamings and substitutions in System F. The only difference is that we define the substitution operator only on types.

```
_[_] : Type (S ▷ τ_s) → Type S → Type S
τ [ τ' ] = sub (single-type_s id_s τ') τ
```

The single-type$_s$ function only introduces a new binding for types and not for arbitrary terms. Because we do not formalize direct semantics for System F_O, only substitutions of types in types are necessary. Type in type substitution appears in the typing rule for type application.

Context

In addition to types and kinds, the existence of overloaded variables is stored inside the context. Overloaded variables act as normal context items. Because overloaded variables themselves do not have a type, but rather multiple types that they can take on, we only need to store their existence in Γ. Thus, similar to type variables, we store kind $\star$ in Γ to denote the existence of an overloaded variable.
The types that an overloaded variable can take on are stored in the form of constraints. Constraints can be introduced to the context by both constraint abstractions and instance expressions.

```
data Ctx : Sorts → Set where
  _▸_ : Ctx S → Cstr S → Ctx S
  -  ...
```

We write $\Gamma \blacktriangleright c$ to pick up a constraint c. Because constraints give an additional meaning to an overloaded variable that is already bound, the index S is not modified.

The lookup function in System F_O is defined analogous to the lookup function in System F and simply ignores constraints stored in the context.

Constraint Solving

The search for constraints in a context is inductively formalized, similar to membership proofs $s \in S$. The subtle difference is that we reference constraints in Γ and not in S. The constraint solving type does need to search in Γ because S does not know about the existence of constraints.

```
data [_]∈_ : Cstr S → Ctx S → Set where
  here : [ (‘ o : τ) ]∈ (Γ ▶ (‘ o : τ))
  under-bind : {I : Term S (item-of s')} →
    [ (‘ o : τ) ]∈ Γ → [ (‘ there o : wk τ) ]∈ (Γ ▶ I)
  under-cstr : [ c ]∈ Γ → [ c ]∈ (Γ ▶ c')
```

The here constructor is analogous to the here constructor of memberships and can be used when the last item in Γ is the desired constraint c.
If the last item in the context is not the desired constraint c, then c must be further inside the context. The constraint can either be behind an item stored in Γ (under-bind) or a constraint (under-cstr). In the case that c is under a binder, the constraint needs to be weakened to align in S with the position that it is resolved for.
We use the constraint solving type inside the type system to resolve the instance for usages of overloaded variables and to implicitly eliminate constraints.

Typing

We only discuss typing rules not already discussed in the System F specification. The typing for overloaded variables results in a type. As a result, the type-of function returns the sort τ_s for the sort o_s in the case of typings.

```
data _⊢_:_ : Ctx S → Term S s → Term S (type-of s) → Set where
  ⊢‘o :
    [ ‘ o : τ ]∈ Γ →
    Γ ⊢ ‘ o : τ
  ⊢λ :
    Γ ▶ c ⊢ e : τ →
    Γ ⊢ λ c ⇒ e : [ c ]⇒ τ
  ⊢∅ :
    Γ ⊢ e : [ ‘ o : τ ]⇒ τ' →
    [ ‘ o : τ ]∈ Γ →
    Γ ⊢ e : τ'
  ⊢decl :
    Γ ▶ ⋆ ⊢ e : wk τ →
    Γ ⊢ decl‘o‘in e : τ
  ⊢inst :
    Γ ⊢ e₂ : τ →
    Γ ▶ (‘ o : τ) ⊢ e₁ : τ' →
    Γ ⊢ inst‘ ‘ o ‘= e₂ ‘in e₁ : τ'
  - ...
```

The rule for overloaded variables $\vdash$`o` says that if we can resolve the constraint $o : \tau$ in Γ, then o can take on type τ.
The rule for constraint abstractions $\vdash\lambda$ appends the constraint c to Γ and thus assumes c to be valid inside the body e. Constraint abstractions result in expressions of the corresponding constraint type $[\ c\ ] \Rightarrow \tau$ that lifts the constraint onto the type level.
Expressions e with constraint type $[\ c\ ] \Rightarrow \tau$' have the constraint implicitly eliminated using the rule $\vdash\oslash$, given c can be resolved in Γ.
The rule $\vdash$decl introduces a new overloaded variable o to e. To introduce o in Γ, we only need to store the information that o exists as overloaded variable. The existence of o is denoted by extending Γ with kind $\star$. Analogous to the type τ' inside the abstraction rule $\vdash\lambda$, the resulting type τ cannot use the introduced overloaded variable and thus is weakened to align in S with $\Gamma \blacktriangleright \star$. In consequence, τ can act as the resulting type of the typing.
An instance for an overloaded variable o is typed using the rule $\vdash$inst. We extend Γ with constraint $o : \tau$ inside e_1, where τ is the type of e_2.

Typing Renaming & Substitution

Typed renamings are identical to typed renamings in System F, except there is an additional case for the weakening by a constraint.

```
data _:_⇒ᵣ_ : Ren S₁ S₂ → Ctx S₁ → Ctx S₂ → Set where
  ⊢drop-cstrᵣ : ∀ {Γ₁ : Ctx S₁} {Γ₂ : Ctx S₂} {τ} {o} →
    ρ : Γ₁ ⇒ᵣ Γ₂ →
    ρ : Γ₁ ⇒ᵣ (Γ₂ ▶ (o : τ))
  - ...
```

A constraint $o : \tau$ can be introduced only to Γ_2 using the $\vdash$drop-cstr$_r$ constructor. Dropping a constraint corresponds to a typed weakening, similar to constructor $\vdash$drop$_r$, but instead of introducing an unused variable we introduce an unused constraint. In consequence, the typed weakening by a constraint $\vdash$wk-cstr$_r$ is defined by $\vdash$drop-cstr$_r$ $\vdash$id$_r$.
Other than in System F, arbitrary substitutions will not be allowed in System F_O. Similar to the substitution operator we restrict typed substitutions in System F_O to substitutions of types in types.

```
data _:_⇒ₛ_ : Sub S₁ S₂ → Ctx S₁ → Ctx S₂ → Set where
  ⊢single-typeₛ : ∀ {Γ₁ : Ctx S₁} {Γ₂ : Ctx S₂} {τ : Type S₂} →
    σ : Γ₁ ⇒ₛ Γ₂ →
    single-typeₛ σ τ : Γ₁ ▶ ⋆ ⇒ₛ Γ₂
  - ...
```

The constructor $\vdash$single-type$_s$ allows to introduce an additional new type variable binder that is substituted with type τ in Γ_1. Thus, the constructor $\vdash$single-type$_s$ complements the single-type$_s$ function.
Constructors $\vdash$id$_s$, $\vdash$ext$_s$, $\vdash$drop$_s$ and $\vdash$drop-cstr$_s$ are not shown. All of them function the same way as their counterparts in typed renamings.
The intuition here is that if we would allow all terms to be introduced using a $\vdash$term$_s$ constructor, then typed substitutions in System F_O would be arbitrary again. The restriction to type in type substitutions simplifies the type preservation proof for the

Dictionary Passing Transform by eliminating cases for non-type terms that would otherwise needed to be proven. In hindsight arbitrary substitutions would not have produced an unreasonable amount of additional work, but the restriction did not have any negative effects, so it remained as is.

5 The Dictionary Passing Transform

5.1 Translation

Sorts

The translation of System F_O sorts to System F sorts only considers sorts that are bindable. The two missing non-bindable sorts c_s and κ_s do not need to be translated. Intuitively there does not even exist a sensible translation for c_s.

```
s⇝s : F^O.Sort var → F.Sort var
s⇝s e_s = e_s
s⇝s o_s = e_s
s⇝s τ_s = τ_s
```

Sorts e_s and τ_s translate to their corresponding counterparts in System F.
The sort o_s translates to e_s. This is because, instead of removing decl expressions as seen in the example at the beginning, we keep them as redundant let expressions that bind a unit value tt. Because decl expressions bind a new overloaded variable in System F_O, removing them would result in a variable binding less in System F. To prevent more complex proofs involving the index S of terms, we keep all decl expressions as redundant let bindings. In consequence, all bindings for overloaded variables translate to normal expression bindings in System F and thus, the sort o_s needs to translate to e_s.
Translating the index S directly is not possible because there might appear additional sorts inside the list after the translation. New sorts must be added for variable bindings introduced by the translation. For example, an inst‘ ‘ $o = e_2$ ‘in e_1 expression does not bind a new variable in e_1, but translates to a let‘x= e_2 ‘in e_1 binding. Hence, S must have an additional entry e_s at the corresponding position to further function as valid index for the translated e_1. Similarly, an additional sort e_s must be appended to S inside the body of translated constraint abstractions. To solve this problem the System F_O context Γ is used to build the translated S. The context stores the relevant information about introduced constraints and thus all positions of new bindings that were not present in System F_O.

```
Γ⇝S : F^O.Ctx F^O.S → F.Sorts
Γ⇝S ∅ = []
Γ⇝S (Γ ▸ c) = Γ⇝S Γ ▷ F.e_s
Γ⇝S {S ▷ s} (Γ ▸ T) = Γ⇝S Γ ▷ s⇝s s
```

The empty context $\emptyset$ corresponds to the empty list [].
For each constraint in Γ an additional sort e_s is appended to S.
If we find that a normal item is stored in the context, then the sort s is directly translated using the function s⇝s.

Variables

Similar to the translation of sort lists S, the translation for variables needs context information.

```
x⇝x : ∀ {Γ : F^O.Ctx F^O.S} →
  F^O.Var F^O.S F^O.s → F.Var (Γ⇝S Γ) (s⇝s F^O.s)
x⇝x {Γ = Γ ▶ τ} (here refl) = here refl
x⇝x {Γ = Γ ▶ τ} (there x) = there (x⇝x x)
x⇝x {Γ = Γ ▶ c} x = there (x⇝x x)
```

If an item is stored in the context we can translate the variable directly.
Whenever a constraint is encountered, x is wrapped in an additional there. This is because the expression that introduced the constraint will translate to an expression with an additional new binding that needs to be respected in System F.
Resolved constraints translate to correct unique expression variables using function o⇝x. We can apply a symmetric argumentation as seen in the translation for variables because the type for resolved constraints [c]∈ Γ preserves the structure of the context perfectly. The subtle difference to x⇝x is that we have the two cases here and under-cstr for constraints, instead of the two cases here and there for normal variables. Futhermore, we only have one case for bindings under-bind in o⇝x, instead of one case for constraints in x⇝x.

```
o⇝x : ∀ {Γ : F^O.Ctx F^O.S} →
  [ ‘ F^O.o : F^O.τ ]∈ Γ → F.Var (Γ⇝S Γ) F.e_s
o⇝x here = here refl
o⇝x (under-bind o:τ∈Γ) = there (o⇝x o:τ∈Γ)
o⇝x (under-cstr o:τ∈Γ) = there (o⇝x o:τ∈Γ)
```

Inside the case here we found the correct instance, now variable.
When we encounter a normal binding in the case under-bind, we wrap the variable in a there constructor to respect the binding.
In the case under-cstr, we again wrap the variable in an additional there that was not present before to respect the new binding introduced by the translation.

Context

The translation of contexts is mostly a direct translation. We only look at the translation of constraints stored in the context.

```
Γ⇝Γ : (Γ : F^O.Ctx F^O.S) → F.Ctx (Γ⇝S Γ)
Γ⇝Γ (Γ ▶ (‘ o : τ)) = (Γ⇝Γ Γ) ▶ τ⇝τ τ
- ...
```

Following the idea from above, constraints $o : \tau$ stored inside Γ translate to normal items in the translated Γ. The item introduced is the translated type τ⇝τ τ that was originally required by the constraint. This is exactly what we want because for each constraint in System F_O there will either be an additional lambda binding in System F that accepts the constraint as higher-order function or a let binding that binds a variable with type τ⇝τ τ.

Renaming & Substitution

Typed renamings in System F_O translate to untyped renamings in System F.

```
⊢ρ⇝ρ : ∀ {ρ : F^O.Ren F^O.S₁ F^O.S₂} {Γ₁ : F^O.Ctx F^O.S₁} {Γ₂ : F^O.Ctx F^O.S₂} →
  ρ F^O.: Γ₁ ⇒ᵣ Γ₂ →
  F.Ren (Γ⇝S Γ₁) (Γ⇝S Γ₂)
⊢ρ⇝ρ (⊢drop-cstrᵣ ⊢ρ) = F.dropᵣ (⊢ρ⇝ρ ⊢ρ)
- ...
```

Because constraints translate to actual bindings, the constructor $\vdash$drop-cstr$_r$ translates to drop$_r$ in System F.
The typed renamings $\vdash$id$_r$, $\vdash$ext$_r$ and $\vdash$drop$_r$ translate to their untyped counterparts.
The translation of typed substitutions to untyped substitutions follows similarly.

```
⊢σ⇝σ : ∀ {σ : F^O.Sub F^O.S₁ F^O.S₂} {Γ₁ : F^O.Ctx F^O.S₁} {Γ₂ : F^O.Ctx F^O.S₂} →
  σ F^O.: Γ₁ ⇒ₛ Γ₂ →
  F.Sub (Γ⇝S Γ₁) (Γ⇝S Γ₂)
⊢σ⇝σ (⊢single-typeₛ {τ = τ} ⊢σ) = F.singleₛ (⊢σ⇝σ ⊢σ) (τ⇝τ τ)
- ...
```

The typed renaming $\vdash$single-type$_s$ translates to its untyped counterpart for arbitrary terms single$_s$.
The cases $\vdash$id$_s$, $\vdash$ext$_s$, $\vdash$drop$_s$ and $\vdash$drop-cstr$_s$ are analogous to the cases for renamings.

Terms

Types and kinds can be translated without typing information using the function T⇝T.
We only look at the untyped translation of types τ⇝τ that is used inside the function T⇝T because the kind ⋆ translates to its direct counterpart in System F.

```
τ⇝τ : ∀ {Γ : F^O.Ctx F^O.S} →
  F^O.Type F^O.S →
  F.Type (Γ⇝S Γ)
τ⇝τ ([ o : τ ]⇒ τ') = τ⇝τ τ ⇒ τ⇝τ τ'
- ...
```

Constraint types [o : τ]⇒ τ' translate to function types $\tau \Rightarrow \tau$'. The translation from constraint types to function types directly corresponds to the translation of constraint abstractions to normal abstractions. The implicitly resolved constraint will be taken as higher-order function argument of type τ⇝τ τ.
All other System F_O types translate to their direct counterparts in System F.
Arbitrary terms can only be translated using typing information. The typing carries information about the instances that were resolved for usages of overloaded variables and the instances that were implicitly resolved for constraints. We only look at the translation of System F_O expressions that do not have a direct counterpart in System F.

```
⊢t⇝t : ∀ {Γ : F^O.Ctx F^O.S} {t : F^O.Term F^O.S F^O.s}
         {T : F^O.Term F^O.S (F^O.type-of F^O.s)} →
  Γ F^O.⊢ t : T →
  F.Term (Γ⇝S Γ) (s⇝s F^O.s)
⊢t⇝t (⊢`o o:τ∈Γ) = ` o⇝x o:τ∈Γ
⊢t⇝t (⊢λ ⊢e) = λ`x→ (⊢t⇝t ⊢e)
⊢t⇝t (⊢⊘ ⊢e o:τ∈Γ) = ⊢t⇝t ⊢e · ` o⇝x o:τ∈Γ
⊢t⇝t (⊢decl ⊢e) = let`x= tt `in ⊢t⇝t ⊢e
⊢t⇝t (⊢inst ⊢e₂ ⊢e₁) = let`x= ⊢t⇝t ⊢e₂ `in ⊢t⇝t ⊢e₁
- ...
```

Typed overloaded variables ⊢`o carry the information $o{:}\tau{\in}\Gamma$ about the instance that was resolved for o. We translate the resolved instance to the corresponding unique variable in System F using the o⇝x function from above.
Typed constraint abstractions ⊢λ translate to normal abstractions with an additional new binding.
An implicitly resolved constraint ⊢⊘ translates to an explicit application that passes the resolved instance $o{:}\tau{\in}\Gamma$ as argument. We again use function o⇝x to translate the resolved instance to the corresponding unique variable.
As discussed, we translate typed declaration expressions ⊢decl to redundant let bindings that bind a unit value.
We translate typed instance expressions ⊢inst to let expressions that introduce an additional binding not present in System $\mathrm{F_O}$.

5.2 Type Preservation

Terms

We first look at the final proof of type preservation for the Dictionary Passing Transform to motivate all necessary lemmas. Type preservation is proven by induction over the typing rules of System $\mathrm{F_O}$. The function ⇝-pres-⊢ produces a typed System F term for an arbitrary typed System $\mathrm{F_O}$ term $\vdash t$. The translated System $\mathrm{F_O}$ term ⊢t⇝t t is typed in the translated context Γ⇝Γ Γ and has the typing result T⇝T T.

```
⇝-pres-⊢ : {Γ : F^O.Ctx F^O.S} {t : F^O.Term F^O.S F^O.s}
             {T : F^O.Term F^O.S (F^O.type-of F^O.s)} →
  (⊢t : Γ F^O.⊢ t : T) →
  (Γ⇝Γ Γ) F.⊢ (⊢t⇝t ⊢t) : (T⇝T T)
⇝-pres-⊢ (⊢`x Γx≡τ) = ⊢`x (⇝-pres-lookup Γx≡τ)
⇝-pres-⊢ (⊢`o o:τ∈Γ) = ⊢`x (⇝-pres-cstr-solve o:τ∈Γ)
⇝-pres-⊢ (⊢let ⊢e₂ ⊢e₁) = ⊢let (⇝-pres-⊢ ⊢e₂)
  (subst (_ F.⊢ ⊢t⇝t ⊢e₁ :_) ⇝-dist-wk-type(⇝-pres-⊢ ⊢e₁))
⇝-pres-⊢ (⊢λ {c = (` o : τ)} ⊢e) = ⊢λ
  (subst (_ F.⊢ ⊢t⇝t ⊢e :_) ⇝-dist-wk-inst-type (⇝-pres-⊢ ⊢e))
⇝-pres-⊢ (⊢⊘ ⊢e o:τ∈Γ) = ⊢· (⇝-pres-⊢ ⊢e) (⊢`x (⇝-pres-cstr-solve o:τ∈Γ))
⇝-pres-⊢ (⊢• {τ = τ} {τ' = τ'} ⊢e) = subst (_ F.⊢ ⊢t⇝t ⊢e • τ⇝τ τ' :_)
  (⇝-dist-τ[τ'] τ' τ) (⊢• (⇝-pres-⊢ ⊢e))
- ...
```

Proof $\Gamma x{\equiv}\tau$ that a variable x has type τ in Γ translates to proof that x⇝x x has type τ⇝τ τ in Γ⇝Γ Γ using lemma ⇝-pres-lookup. The lemma produces an equality proof

of type F.lookup (Γ⇝Γ Γ) (x⇝x x) ≡ (τ⇝τ τ) when given an equality proof F^O.lookup Γ x ≡ τ . With the lemma ⇝-pres-lookup the typing rule ⊢'x can be translated to the typing rule for variables in System F.
Similarly, the lemma ⇝-pres-cstr-solve translates the proof $o{:}\tau{\in}\Gamma$ that an instance o : τ was resolved for an overloaded variable o to proof that the unique variable o⇝x $o{:}\tau{\in}\Gamma$ has type τ⇝τ τ in Γ⇝Γ Γ: F.lookup (Γ⇝Γ Γ) (o⇝x $o{:}\tau{\in}\Gamma$) ≡ (τ⇝τ τ). Using lemma ⇝-pres-cstr-solve the typing rule for overloaded variables ⊢'o can be translated to the typing rule for normal variables ⊢'x.
Typed let bindings ⊢let $\vdash e_2$ $\vdash e_1$ translate to typed let bindings in System F. The typing rule $\vdash e_2$ is translated directly using the induction hypothesis. Because the expression e_1 results in type wk τ' inside typing rule $\vdash e_1$, proof is needed that τ' weakened in System F_O and translated to System F is equivalent to the weakening of the translated τ' in System F. Lemma ⇝-dist-wk-type produces the required equality proof of type τ⇝τ $\{\Gamma = \Gamma \blacktriangleright T\}$ (F^O.wk τ') ≡ F.wk (τ⇝τ τ'). On the left hand side of the equality we need to specify that the otherwise implicit argument $\Gamma = \Gamma$ needs to be an extended Γ in that case. We then substitute the equivalence into the translation of the typing rule $\vdash e_1$.
Typed constraint abstractions ⊢λ translate to normal abstractions in System F. Inside the typing $\vdash e$, the result type τ of the body e does not need to be weakened because the constraint abstraction only introduced a constraint and no actual binding to context Γ. After the translation the former constraint will be bound by a binding and thus a new item in Γ⇝Γ Γ will exist. To ignore the binding, τ is weakened in the abstraction rule ⊢λ. Lemma ⇝-dist-wk-inst-type proves that translating τ in Γ extended by a constraint is equivalent to weakening τ after the translation: τ⇝τ $\{\Gamma = \Gamma \blacktriangleright (\text{`}\ o : \tau')\}$ τ ≡ F.wk (τ⇝τ τ). The lemma follows because the constraint translates to an actual binding and consequently, both sides of the equivalence have an additional expression binding that τ does not care about.
Implicitly resolved constraints ⊢∅ carry the information $o{:}\tau{\in}\Gamma$ about the instance that was resolved. In System F the former constraint is explicitly passed as variable pointing to the correct translated instance. Thus, the typing ⊢∅ results in typed application ⊢·. We apply the correct equality proof to the typing rule ⊢'x using lemma ⇝-pres-cstr-solve.
The type application rule ⊢• contains type in type substitution. Hence, we need proof that it is irrelevant, if τ' is substituted into τ and then translated or both τ and τ' are translated and substitution is applied in System F. Using lemma ⇝-dist-τ[τ'] of type (τ⇝τ $\{\Gamma = \Gamma \blacktriangleright \star\}$ τ' F.[τ⇝τ τ]) ≡ τ⇝τ (τ' F^O.[τ]) we can substitute the equivalence into the System F typing rule ⊢• (⇝-pres-⊢ $\vdash e$).
The translation of ⊢⊤, ⊢λ, ⊢·, ⊢decl and ⊢inst is either a direct translation or uses similar ideas and no other lemmas than the ones discussed.

Renaming

Both ⇝-dist-wk-type and ⇝-dist-wk-inst-type directly follow from a more general lemma ⇝-dist-ren-type for arbitrary renamings. The lemma ⇝-dist-ren-type proves that translating both the typed renaming $\vdash \rho$ and type τ and then applying the renaming in System F is equivalent to applying the renaming ρ in System F_O and then translating the renamed type τ. The lemma can be proven by induction over System F_O types τ.

```
⇝-dist-ren-type : {ρ : F^O.Ren F^O.S₁ F^O.S₂}
                  {Γ₁ : F^O.Ctx F^O.S₁} {Γ₂ : F^O.Ctx F^O.S₂} →
  (⊢ρ : ρ F^O.: Γ₁ ⇒ᵣ Γ₂) →
  (τ : F^O.Type F^O.S₁) →
  F.ren (⊢ρ⇝ρ ⊢ρ) (τ⇝τ τ) ≡ τ⇝τ (F^O.ren ρ τ)
⇝-dist-ren-type ⊢ρ (` x) = cong `_ (⇝-dist-ren-var-type ⊢ρ x)
⇝-dist-ren-type ⊢ρ ([ ` o : τ ]⇒ τ') = cong₂ _⇒_
  (⇝-dist-ren-type ⊢ρ τ) (⇝-dist-ren-type ⊢ρ τ')
- ...
```

The case for type variables needs an additional lemma ⇝-dist-ren-var-type specifically for type variables. Lemma ⇝-dist-ren-var-type proves an analogous statement, but for type variables applied to a renaming: (⊢ρ⇝ρ ⊢ρ) _ (x⇝x x) ≡ x⇝x (ρ x). This statement can be proven via straight forward induction over typed System F_O renamings $\vdash\rho$.
All other cases follow directly from the induction hypothesis. The only small exception is the constraint type, where we need to respect that it translates to a function type.

Substitution

Similar to renamings, the lemma for single substitution on types ⇝-dist-τ[τ'] follows from a more general lemma about type in type substitutions ⇝-dist-sub-type. The lemma ⇝-dist-sub-type also follows by straight forward induction over System F_O types, except the case for type variables. Other than with renamings, the cases for lemma ⇝-dist-sub-var-type do not follow directly from the induction hypothesis. To understand why, we at look at the case $\vdash ext_s$.

```
⇝-dist-sub-var-type : {σ : F^O.Sub F^O.S₁ F^O.S₂}
                      {Γ₁ : F^O.Ctx F^O.S₁} {Γ₂ : F^O.Ctx F^O.S₂} →
  (⊢σ : σ F^O.: Γ₁ ⇒ₛ Γ₂) →
  (x : F^O.Var F^O.S₁ τₛ) →
  F.sub (⊢σ⇝σ ⊢σ) (` x⇝x x) ≡ τ⇝τ (F^O.sub σ (` x))
⇝-dist-sub-var-type (⊢ext_s ⊢σ) (here refl) = refl
⇝-dist-sub-var-type (⊢ext_s {σ = σ} ⊢σ) (there x) = trans
  (cong F.wk (⇝-dist-sub-var-type ⊢σ x)) (⇝-dist-ren-type F^O.⊢wkᵣ (σ x))
```

The case $\vdash ext_s$ is proven via a case split on variables, similar to how ext_s is defined. The base case holds by definition. In the induction case we weaken both sides of the equality that results from the outer induction hypothesis. We then combine the weakened induction hypothesis with proof that weakenings preserve the translation using transitivity. The intuition here is that we need the renaming lemma ⇝-dist-ren-type applied to the typed weakening $\vdash wk_r$ because ext_s is defined by weakening the terms that result of the substitution σ being applied to variables x.
Both $\vdash id_s$ and $\vdash single\text{-}type_s$ follow directly from the induction hypothesis. The cases for $\vdash drop_s$ and $\vdash drop\text{-}cstr_s$ are similar to the case $\vdash ext_s$ and also align with their corresponding definition.

Variables

We first look at the proof for lemma ⇝-pres-lookup. The lemma is proven via induction over the System F_O context Γ.

```
⇝-pres-lookup : ∀ {Γ : F^O.Ctx F^O.S} {τ : F^O.Type F^O.S} {x : F^O.Var F^O.S e_s} →
  F^O.lookup Γ x ≡ τ →
  F.lookup (Γ⇝Γ Γ) (x⇝x x) ≡ (τ⇝τ τ)
⇝-pres-lookup {Γ = Γ ▶ τ} {x = here refl} refl = ⇝-dist-ren-type F^O.⊢wk_r τ
⇝-pres-lookup {Γ = Γ ▶ _} {τ'} {x = there x'} refl = trans
  (cong F.wk (⇝-pres-lookup {x = x'} refl))
  (⇝-dist-ren-type F^O.⊢wk_r (F^O.lookup Γ x'))
- ...
```

As an example we will look at the case $\Gamma \blacktriangleright \tau$. The case is proven via a case split on variables. The prove follows the same reasoning as the $\vdash ext_s$ case for substitutions above. Because the function lookup weakens the type τ that is looked up in Γ in both cases here and there, both use lemma ⇝-dist-ren-type applied to the typed weakening $\vdash wk_r$ to account for the weakening.
The case $\Gamma \blacktriangleright c$ follows analogously, but uses $\vdash$wk-cstr$_r$ applied to the induction hypothesis instead of $\vdash wk_r$. Furthermore, the case $\emptyset$ is impossible because there must be a context item if a variable exists.
The lemma ⇝-pres-cstr-solve can be proven via induction over the type for resolved constraints $[\ c\ ] \in \Gamma$. The lemma follows symmetrically to the lemma ⇝-pres-lookup because the type for resolved constraints preserves the structure of Γ perfectly.

```
⇝-pres-cstr-solve : ∀ {Γ : F^O.Ctx F^O.S} →
  (o:τ∈Γ : [ ' F^O.o : F^O.τ ]∈ Γ) →
  F.lookup (Γ⇝Γ Γ) (o⇝x o:τ∈Γ) ≡ (τ⇝τ F^O.τ)
```

Similar to the case split on variables with constructors here and there inside the case $\Gamma \blacktriangleright \tau$ of lemma ⇝-pres-lookup, we have the two cases here and under-cstr in lemma ⇝-pres-cstr-solve. Both cases work similarly to cases in lemma ⇝-pres-lookup, except that they use $\vdash$wk-cstr$_r$ instead of $\vdash wk_r$.
Furthermore, we have the case under-bind that works similar to the case $\Gamma \blacktriangleright c$ and always uses the induction hypothesis applied to $\vdash wk_r$ because we search for a former constraint and not a variable.

6 Discussion and Conclusion

6.1 Hindley Milner with Overloading

In this scenario the source language for the Dictionary Passing Transform would be an extended Hindley-Milner [8] based system HM_O and the target language would be Hindley-Milner. The Hindley-Milner system is a restricted form of System F that allows for full type inference. Similarly, HM_O would be a restricted form of System F_O with support for full type inference.
Formalizing Hindley-Milner would require two new sorts, m_s and p_s for mono and poly types, in favour of the sort for arbitrary types τ_s. Poly types can include quantification over type variables while mono types consist only of primitive types and type variables.

Usually all language constructs are restricted to mono types, except let bound variables. Hence, polymorphism in Hindley-Milner is also called let polymorphism. As a result, constraints must have the form $o : m$, where m is a mono type. To separate the type logic from the expression logic in Hindley-Milner fashion, we would need to embed constraints into explicit type annotations of instances, instead of introducing them on the expression level. The explicit type annotation for instances would allow poly types because instance expressions translate to let bindings after all. But instances would need to be restricted as well. For each overloaded variable o, all instances would need to differ in the type of their first argument.
With these two restrictions full type inference for instances and overloaded variables should be preserved. The inference algorithm would treat instance expressions similar to let bindings and could infer the type of an overloaded identifier via the type of the first argument applied. For now it remains unclear, if the inference algorithm can be extended to work for arbitrary mono type constraints and how constraints should be handled by the inference algorithm in general.

6.2 Proving Semantic Preservation

For now System F_O does not have direct semantics formalized. In section 2.3 we already discussed that correct semantics are already implicitly given by the translation, but it could also be interesting to investigate direct semantics on the System F_O syntax.
Semantics for System F_O would need to be typed semantics because applications $` o \cdot e_1 \cdot .. \cdot e_n$ need type information to reduce properly. The correct instance for $o$ needs to be substituted based on the types of the arguments $e_1 .. e_n$. More specifically, we could introduce a reduction rule β-o-* that reduces $` o \cdot e_1 \cdot .. \cdot e_n$ to $e \cdot e_1 \cdot .. \cdot e_n$, where e is the resolved instance based on the types of $e_1 .. e_n$. The drawback would be that partial application to overloaded variables would not be possible. Alternately, we could apply the restriction mentioned above that restricts all instances for an overloaded variable o to differ in the type of their first argument. In consequence, the resolved instance for o in a single application step $` o \cdot e$ would be decidable.
Let $\vdash e \hookrightarrow \vdash e'$ be a typed small step semantic for System F_O. We would need to prove something similar to: If $\vdash e \hookrightarrow \vdash e'$ then $\exists\ [\ e''\ ]\ (\vdash e \hookrightarrow e' \rightsquigarrow e \hookrightarrow e' \vdash e \hookrightarrow^* e'') \times (\vdash e \hookrightarrow e' \rightsquigarrow e \hookrightarrow e' \vdash e' \hookrightarrow^* e'')$, where $\vdash e \hookrightarrow e' \rightsquigarrow e \hookrightarrow e'$ translates typed System F_O reductions to a untyped System F reductions. Instead of translating reduction steps directly, we prove that both translated $\vdash e$ and translated $\vdash e'$ reduce to a System F expression e'' using finite many reduction steps. This more general formulation is needed because there might be more reduction steps in the translated System F expression than in the System F_O expression. For example, an implicitly resolved constraint in System F_O needs to be explicitly passed as argument using an additional application step in System F. For now it remains unclear if semantic preservation can be proven by induction over typed semantic rules or if logical relations are needed [1].

6.3 Related Work

The ideas for the required restrictions to preserve the inference algorithm in section 6.1 originate from System O [9]. System O is a language extension to the Hindley-Milner system. In contrast to System F_O constraints are not introduced on the expression level and instead are introduced via explicit type annotations of instances as part of forall types.

For instance, the valid System F_O and HM_O type $\forall\alpha.\ \forall\beta.\ [a : \alpha \to \alpha \to \alpha] \Rightarrow [b : \beta \to \beta \to \beta] \Rightarrow ..$ would be expressed as $\forall\alpha.\ (a : \alpha \to \alpha \to \alpha) \Rightarrow \forall\beta.\ (b : \beta \to \beta \to \beta) \Rightarrow ..$ in System O. Inside the System O type, we only introduce one constraint per type variable, but a list of constraints would be allowed. The part about the inference algorithm that remained unclear in section 6.1 is solved in System O by restricting constraints to begin their type with the type variable that is bound by the quantifier that they are part of. In HM_O such a connection would not exist.
Originally the plan was to formalize System O in Agda, but multiple issues arose in the type preservation proof. First, because we have a list of n constraints in each forall type, translating them results in n new lambda bindings in one induction step. While the problem above can be handled, another problem complicated the proof of type preservation via induction immensely. The translation of System O types must pull out forall quantifiers, because translating constraints directly to higher-order functions would break the rule that function types are only allowed to be built from mono types. For instance, the translated System O type from above should not be $\forall\alpha.\ (\alpha \to \alpha \to \alpha) \to \forall\beta.\ (\beta \to \beta \to \beta) \to ..$, but rather $\forall\alpha.\ \forall\beta.\ (\alpha \to \alpha \to \alpha) \to (\beta \to \beta \to \beta) \to \lrcorner\ ..$ to be a valid Hindley-Milner type. Including the additional transform on types complicates the type preservation proof immensely, because the transform affects the type of the next n expressions and thus straight forward induction did not work out.
Besides System O, there exist other formalizations that are closer related to typeclasses in Haskell [10] [4]. A more general approach to constraint types is presented by the theory of qualified types [6].

6.4 Conclusion

We have formalized both System F and System F_O in Agda. In the process, we explored the technique of using an intrinsically scoped and sorted data type to represent syntaxes. The essence of System F_O was to act as a core calculus that captures the idea of overloading and type constraints. We formalized the Dictionary Passing Transform between System F and System F_O. Furthermore, we proved the System F formalization to be sound and the Dictionary Passing Transform from System F_O to System F to be type preserving. The full formalization of the Dictionary Passing Transform, System F_O and System F can be found as Agda code files [2].
One trick used in the formalization was to introduce constraints individually using constraint abstractions. As a result, we were able to translate constraint abstractions directly to lambda abstractions in System F. Another trick that we used was to preserve the structure of the context inside the type for resolved constraints. In consequence, the translation of resolved constraints to Debruijn variables was straight forward because the position of new bindings introduced by the translation was perfectly known.
A reasonable next step would be to define direct semantics for System F_O and prove semantic preservation for the Dictionary Passing Transform.

[2] Formalizations and proofs as Agda code files: `https://github.com/Mari-W/System-Fo/tree/main/proofs`

References

1. Abel, A., Allais, G., Hameer, A., Pientka, B., Momigliano, A., Schäfer, S. & Stark, K. POPLMark reloaded: Mechanizing proofs by logical relations. *Journal Of Functional Programming*. **29** pp. e19 (2019), http://dx.doi.org/10.1017/S0956796819000170
2. Barendregt, H. Introduction to generalized type systems. *Journal Of Functional Programming*. **1**, 125-154 (1991), https://doi.org/10.1017%2Fs0956796800020025
3. Bove, A., Dybjer, P., Norell, U. (2009). A Brief Overview of Agda – A Functional Language with Dependent Types. In: Berghofer, S., Nipkow, T., Urban, C., Wenzel, M. (eds) Theorem Proving in Higher Order Logics. TPHOLs 2009. Lecture Notes in Computer Science, vol 5674. Springer, Berlin, Heidelberg. https://doi.org/10.1007/978-3-642-03359-9_6
4. Cordelia V. Hall, Kevin Hammond, Simon L. Peyton Jones, and Philip L. Wadler. 1996. Type classes in Haskell. ACM Trans. Program. Lang. Syst. 18, 2 (March 1996), 109–138. https://doi.org/10.1145/227699.227700
5. Chapman, J., Kireev, R., Nester, C. & Wadler, P. System F in Agda, for Fun and Profit. (2019,10)
6. Jones, M.P. (1992). A theory of qualified types. In: Krieg-Brückner, B. (eds) ESOP '92. ESOP 1992. Lecture Notes in Computer Science, vol 582. Springer, Berlin, Heidelberg. https://doi.org/10.1007/3-540-55253-7_17
7. Martin-Löf, P. & Sambin, G. Intuitionistic Type Theory. (Bibliopolis, 1984), https://www.cse.chalmers.se/~peterd/papers/MartinL%C3%B6f1984.pdf
8. Milner, R. A theory of type polymorphism in programming. *Journal Of Computer And System Sciences*. **17**, 348-375 (1978), https://www.sciencedirect.com/science/article/pii/0022000078900144
9. Odersky, M., Wadler, P. & Wehr, M. A Second Look at Overloading. *Proceedings Of The Seventh International Conference On Functional Programming Languages And Computer Architecture*. pp. 135-146 (1995), https://doi.org/10.1145/224164.224195
10. P. Wadler and S. Blott. 1989. How to make ad-hoc polymorphism less ad hoc. In Proceedings of the 16th ACM SIGPLAN-SIGACT symposium on Principles of programming languages (POPL '89). Association for Computing Machinery, New York, NY, USA, 60–76. https://doi.org/10.1145/75277.75283

Declaration

I hereby declare, that I am the sole author and composer of my thesis and that no other sources or learning aids, other than those listed, have been used. Furthermore, I declare that I have acknowledged the work of others by providing detailed references of said work.
I also hereby declare that my thesis has not been prepared for another examination or assignment, either in its entirety or excerpts thereof.

Freiburg i. Br, 27.03.2023 M. Weidner

Place, Date Signature